

SAS Technical News

Summer 2006

*For Higher
Customer Satisfaction,
We Bridge
the SAS System
Between
Customer's World.*

CONTENTS

- 1 分析のためにSASプロシジャを使いこなす
- 10 Q&A
- 14 新刊マニュアルのお知らせ
- 15 SASトレーニングのお知らせ
- 15 SAS認定試験のお知らせ
- 16 最新リリース情報
- 16 SAS Technical News送付についてのご案内

特集

分析のために SASプロシジャを 使いこなす

1. はじめに

統計解析のためのソフトウェアとして誕生したSASには、現在でも分析用途のために用意されている機能が数多く存在します。そのため、SASに初めて触れる方々にとって、自分の行ないたい分析をどのようなプログラムで実現できるか、それをマニュアルなどから見つけることが大変だということ意見をいただくことがあります。

また、長い間SASをご利用いただいている方々でも、存在は知っているが実際に使用したことがない機能もあることでしょう。

今号の特集では、主にBase SASに含まれる分析系のプロシジャでどのような処理を行なうことが可能であるか、その概略を紹介します。

2. 言葉の準備

まず最初に、本稿で頻繁に使用する「プロシジャ」「ステートメント」「オプション」という3つのSAS用語を定義しておきましょう。ここでは、データの並べ替え(ソート)を行なうSORTプロシジャを例としてとりあげます。

SORTプロシジャの例

```
PROC SORT DATA=SortData OUT=OutData NOEQUALS;
  BY DESCENDING group; /* 変数groupの値を基準に並べ替え */
RUN;
```

● プロシジャ

PROC ~ という記述で始まり、通常はRUN,QUITまでを、1つの「プロシジャステップ」と呼びます。これは、特定の処理を実現するためのプログラムです。

● ステートメント

1つのステートメントは、セミコロン";"で終了します。

上記の例には、PROC SORTステートメント、BYステートメント、およびRUNステートメントの計3つが存在します。

プロシジャは、複数のステートメントの集合体と考えることができます。

● オプション

オプションは各ステートメントの中で指定し、その処理を行なう際の細かい設定を与えるために使用します。上記の例では、PROC SORTステートメントに、DATA=、OUT=、NOEQUALS、およびBYステートメントにDESCENDINGの各オプションが指定されています。DATA=オプションは、ソートを行なうSASデータセット名を、OUT=オプションは出力先のSASデータセット名を指定するものです。また、NOEQUALSオプションは、並べ替え処理後は同じキーの中では元データにおける相対的な並び順を必ずしも保持しなくてもよいことを指定する機能です。一方、BYステートメントにおけるDESCENDINGオプションは、並び順をデフォルトとは逆にする、具体的には「大きな値」から「小さな値」へ並べるオプションです。SortData、OutDataというSASデータセット名と変数名groupについては、オプションとは呼びません。

3. 共通の機能

以降でご紹介するプロシジャには、共通して使用できるステートメントやオプションがあります。

● BYステートメント

このステートメントで変数を指定すると、その変数値ごとにグループ化されて、分析が行なわれます。

● FREQステートメント、WEIGHTステートメント

度数や重みを表す変数がデータに存在することがあります。SASの多くの分析系プロシジャでは、それらを扱うFREQステートメントとWEIGHTステートメントが使用できます。以降でご紹介するプロシジャについては、RANKプロシジャ以外で利用できます。

● ODS OUTPUTステートメント

それぞれのプロシジャでは、計算結果をSASデータセットとして出力するためのステートメントやオプションが存在しますが、アウトプットに出力される内容はODS OUTPUTステートメントを利用してSASデータセット化することも可能です。詳細については、下記の弊社WEBページをご参照ください。

SAS/STATのプロシジャの結果をデータセットに出力するには
<http://www.sas.com/japan/service/technical/faq/list/body/stat037.html>

4. 分析の説明

4.1. 平均、標準偏差、パーセント点など、基本的な統計量の計算
 平均や標準偏差、また代表的なパーセント点などの基本統計量は、Base SASのMEANSプロシジャで計算できます。最新のSAS 9.1.3では、以下の統計量を算出することができます。

統計量	PROC MEANS ステートメントで指定するキーワード	備考
非欠損値のオブザベーション数	N	
欠損値のオブザベーション数	NMISS	
重み変数の合計	SUMWGT	WEIGHTステートメントで重み変数を指定時のみ有効
最小値	MIN	
最大値	MAX	
範囲	RANGE	最大値-最小値
合計	SUM	
平均	MEAN	
標準偏差	STDDEV または STD	
分散	VAR	
変動係数	CV	

歪度 (ゆがみ)	SKEWNESS または SKEW	
尖度 (とがり)	KURTOSIS または KURT	
修正済平方和	CSS	数値から平均を引いたものを2乗して足しあげる
無修正平方和	USS	数値を2乗して足しあげる
標準誤差	STDERR	
平均に対する両側信頼限界	CLM	両側検定に対応(*)
平均に対する上側信頼限界	UCLM	片側検定に対応(*)
平均に対する下側信頼限界	LCLM	片側検定に対応(*)
t値	T	母平均=0を帰無仮説とした両側検定におけるt統計量
t値に対応するp値	PROBT	上記の検定におけるp値
1パーセント点	P1	
5パーセント点	P5	
10パーセント点	P10	
第1四分位点 (25パーセント点, Q1)	Q1 または P25	
中央値 (メディアン)	MEDIAN または P50	
第3四分位点 (75パーセント点, Q3)	Q3 または P75	
90パーセント点	P90	
95パーセント点	P95	
99パーセント点	P99	
四分位範囲	QRANGE	Q3-Q1

(*)UCLMとLCLMを同時に使用するとCLMを指定することと同等。
 また、信頼水準は、ALPHA=オプションで設定できる。デフォルト設定は、ALPHA=0.05。

PROC MEANSステートメントで、上の表における2列目のキーワードを指定してください。たとえば、各変数の平均、中央値、および合計をアウトプットへ出力するためには、それぞれに対応するキーワードMEAN、MEDIAN、SUMを利用します。

MEANSプロシジャのプログラム例

```

/* サンプルデータの作成 */
DATA MeansData1;
    INPUT x y;
DATA LINES;
2 5
4 6
1 9
3 12
;
RUN;

/* MEANSプロシジャ */
PROC MEANS DATA=MeansData1 MEAN MEDIAN SUM;
    VAR x y;
RUN;
    
```

MEANSプロシジャの出力例

MEANS プロシジャ			
変数	平均	中央値	合計
x	2.5000000	2.5000000	10.0000000
y	8.0000000	7.5000000	32.0000000

もしMEANやMEDIANなどのキーワードの指定を行わずにMEANSプロシジャを実行すると、非欠損値のオブザベーション数、最小値、最大値、平均、および標準偏差が自動的に出力されます。

MEANSプロシジャでは、CLASSステートメントで指定した変数の値に応じて、グループごとに計算を行なうことができます。次のプログラムでは、1と2の値をとる変数group1と、a、b、cの3つの文字値をとる変数group2をサンプルデータの中に用意し、6通りの組み合わせごとに平均などを計算しています。サンプルデータとアウトプットへの出力結果を確認してみましょう。

CLASSステートメントを使用したMEANSプロシジャの例

```

/* サンプルデータの作成 */
DATA MeansData2;
  DO i=1 to 100;
    DO group1=1 TO 2;
      DO group2="a", "b", "c";
        x=group1*10+RANNOR(123);
        y=(group2="a")*5+(group2="b")*10+RANNOR(123);
        z=group1*5+RANNOR(123);
        OUTPUT;
      END;
    END;
  END;
  DROP i;
RUN;
/* 先頭の10オブザベーションを確認のために出力する */
PROC PRINT DATA=MeansData2 (OBS=10);
RUN;
/* CLASSステートメントを使用したMEANSプロシジャ */
PROC MEANS DATA=MeansData2;
  CLASS group1 group2; /* 2つの変数値によってグループ分け */
  VAR x y z;
RUN;

```

MEANSプロシジャで得られた結果をSASデータセットとして出力するには、OUTPUTステートメントを使用します。次のプログラムは、グループごとに最大値を求めて、SASデータセットMeansOut1へ出力するものです。PROC MEANSステートメントのNOPRINTオプションは、アウトプットへの出力を抑制するために使用しています。

OUTPUTステートメントを使用したMEANSプロシジャの例

```

/* OUTPUTステートメントを使用したMEANSプロシジャ */
PROC MEANS DATA=MeansData2 NOPRINT;
  CLASS group1 group2;
  VAR x y z;
  OUTPUT OUT=MeansOut1 MAX=Max_x Max_y Max_z;
RUN;

```

その他、MEANSプロシジャでは、欠損値の扱いに関するオプション、CLASSステートメントを使用したときの出力形態をカスタマイズするオプションや、処理のパフォーマンスを調整するなど、数多くの機能が用意されています。

4.2. 一変数に対するより詳細な解析

Base SASには、UNIVARIATEプロシジャというものもあります。MEANSプロシジャと比べると、大きなデータを扱うときには一般に処理時間がかかりますが、より詳細な分析を行なうときに極めて有用です。

基本的な指定方法はMEANSプロシジャと同じです。UNIVARIATEプロシジャでは、特にオプションを指定しなくても数多くの情報・結果が出力されます。また、本稿ではご紹介しませんが、統計量をSASデータセットへ出力するOUTPUTステートメントや、分類変数ごとに分析するCLASSステートメントが用意されており、基本的にMEANSプロシジャに存在するものと同程度以上の機能を持ちます。

なお、SAS Technical News Spring 2006の特集「SASによるデータ解析の第一歩」の「5. データを眺める、要約する、その2」では、出力される内容の概略が記載されています。併せてそちらもご参考ください。

UNIVARIATEプロシジャの基本的な使用例

```

PROC UNIVARIATE DATA=UnivariateData1;
  VAR x y z;
RUN;

```

UNIVARIATEプロシジャは、他にも様々な統計量を算出します。平均は、データの中心部から大きく離れた数値（いわゆる外れ値）が存在すると、その値から大きな影響を受けてしまいます。そのため、データの属性を表す指標として適切でないことがあります。この点に対する1つの対処策として、Winsor（ウィンザー）化平均やトリム平均（刈り込み平均）といったロバスト（頑健）な平均の指標が提案されています。それぞれWINZORED=、またはTRIMMED=オプションをPROC UNIVARIATEステートメントで指定すると出力されます。なお、=の右には、データの端についてウィンザー化する、またはトリムする割合を指定します。

一方、ROBUSTSCALEオプションは、尺度に対してロバストな統計量を算出します。このオプションをPROC UNIVARIATEステートメントで指定すると、四分位範囲（これはデフォルトでも出力されています）、Gini（ジニ）の平均差、MAD (median absolute deviation from median, 中央値からの絶対偏差の中央値) といった統計量と、正規母集団が仮定されるとき理論値が出力されます。

WINSORED=、およびROBUSTSCALEオプションの指定例

```

PROC UNIVARIATE DATA=UnivariateData1
  WINSORIZED=0.2 ROBUSTSCALE;
  VAR x y z;
RUN;

```

UNIVARIATEプロシジャに備わる他の重要な機能として、データの視覚化とパラメトリック分布の当てはめがあります。データの視覚化に関しては、ヒストグラム（柱状グラフ、histogram）、確率プロット（probability plot）、およびQ-Qプロット（quantile-quantile plot）を描くことができます。それぞれ、HISTOGRAM、PROBPLOT、QQPLOTステートメントを使用してください。

HISTOGRAMステートメントの使用例はSAS Technical News Spring 2006にも記載されていますが、より高度な機能として分布への当てはまり具合を描画することができます。具体的には、ベータ（beta）、指数（exponential）、ガンマ（gamma）、対数正規（lognormal）、正規（normal）、およびワイブル（Weibull）分布への当てはめに対応しており、パラメータの推定結果をアウトプットへ出力し、同時にヒストグラムに理論分布の曲線を重ねあわせることができます。また、核密度推定（カーネル密度推定、kernel density estimation）に基づいて滑らかな分布を描く手法も用意されています。

次の例は、平均が10で標準偏差が3の正規分布に従う乱数列を生成し、ヒストグラム、正規分布への当てはめ曲線（実線）、および核密度推定に基づく曲線（点線）を描くものです。HISTOGRAMステートメントのオプションとして、各分布の名前に由来するNORMALとKERNELと、グラフ表示をカスタマイズするいくつかのオプションを使用しています。

UNIVARIATEプロシジャのHISTOGRAMステートメントの例

```

/* サンプルデータの作成 */
DATA UnivariateData2;
  DO i=1 TO 200;
    x=RANNOR(123)*3+10;
    OUTPUT;
  END;
  DROP i;
RUN;

/* UNIVARIATEプロシジャのHISTOGRAMステートメント */
PROC UNIVARIATE DATA=UnivariateData2;
  HISTOGRAM x / NORMAL(W=6 L=1 COLOR=GRAY00)
  KERNEL(W=6 L=3 COLOR=GRAY77)
  FONT=Arial HEIGHT=4 CFILL=GRAYDD ;
RUN;

```




図1.HISTOGRAM出力グラフの例

確率プロットやQ-Qプロットを描くときも、前述のHISTOGRAMステートメントと同様の指定となります。

UNIVARIATEプロシジャのPROBPLOTステートメントの例

```

/* UNIVARIATEプロシジャのPROBPLOTステートメント */
PROC UNIVARIATE DATA=UnivariateData2;
  PROBPLOT x / NORMAL(MU=EST SIGMA=EST
                      COLOR=GRAY99 L=3 W=4)
  FONT=Arial HEIGHT=4;
RUN;

```




図2.PROBPLOT出力グラフの例

4.3. データの順位付け

データに対して順位付けを行なうためには、Base SASのRANKプロシジャを利用することが一般的です。次のプログラムは、変数x、およびyについて値の小さな方から大きな方へ順位を1、2、3、...とつけます。PROC RANKステートメントのOUT=オプションで指定したSASデータセットへ結果が出力されます。また、RANKSステートメントを使用すると、そこで指定した変数に順位が格納されます。

RANKプロシジャの使用例

```

/* サンプルデータの作成 */
DATA RankData1;
  INPUT x y;
  DATALINES;
  1.3 1.9
  3.6 2.3
  4.5 9.4
  2.9 1.9
  9.3 1.5
  ;
RUN;

/* RANKプロシジャの使用例 */
/* 変数rank_x、rank_yにそれぞれx、yの順位が入ります */
PROC RANK DATA=RankData1 OUT=RankOut1;
  VAR x y;
  RANKS rank_x rank_y;
RUN;

/* PRINTプロシジャでデータを確認 */
PROC PRINT DATA=RankOut1;
RUN;

```

出力例

OBS	x	y	rank_x	rank_y
1	1.3	1.9	1	2.5
2	3.6	2.3	3	4.0
3	4.5	9.4	4	5.0
4	2.9	1.9	2	2.5
5	9.3	1.5	5	1.0

もしRANKSステートメントを使用しなかった場合には、元の変数の値が算出された順位で置き換えられます。

また、値の大きな方から小さな方へ順位付けを逆行しない、値を1、2、3、、、とつける場合には、DESCENDINGオプションをPROC RANKステートメントで指定します。

DESCENDINGオプションの使用例

```
PROC RANK DATA=RankData1 DESCENDING OUT=RankOut2;
  VAR x y;
  RANKS rank_x rank_y;
RUN;
/* PRINTプロシジャでデータを確認 */
PROC PRINT DATA=RankOut2;
RUN;
```

出力例

OBS	x	y	rank_x	rank_y
1	1.3	1.9	5	3.5
2	3.6	2.3	3	2.0
3	4.5	9.4	2	1.0
4	2.9	1.9	4	3.5
5	9.3	1.5	1	5.0

上記の出力では、タイデータが存在します (y=1.9)。それらに対しては、デフォルトでは平均の順位2.5が割り当てられています。タイデータに対する順位付けの方法をコントロールするためには、TIES=オプションを利用してください。デフォルトでは、TIES=MEANという設定が与えられていますが、この順位を低い順位2にあわせるにはTIES=LOWを、また高い順位3にあわせるにはTIES=HIGHをPROC RANKステートメントで指定します。実際にこれらのオプションを指定して確認してみましょう。

TIES=オプションでの順位付け (TIES=HIGH) 例

```
PROC RANK DATA=RankData1 TIES=HIGH OUT=RankOut3;
```

順位付けの結果

y	rank_y	
1.9	3	<--- 順位は3
2.3	4	
9.4	5	
1.9	3	<--- 順位は3
1.5	1	

TIES=オプションでの順位付け (TIES=LOW) 例

```
PROC RANK DATA=RankData1 TIES=LOW OUT=RankOut4;
```

順位付けの結果

y	rank_y	
1.9	2	<--- 順位は2
2.3	4	
9.4	5	
1.9	2	<--- 順位は2
1.5	1	

RANKプロシジャの重要な機能として、データをほぼ同じ大きさのグループに分割できることがあります。PROC RANKステートメントのGROUP=オプションで正の整数を指定すると、VARステートメントで指定した変数の値に応じて、その個数個のグループにおおよそ分割します。次のプログラムではGROUP=10と指定しており、変数xの値で10のグループに分けています。この機能は、いわゆる「デシル分析」の際に有用です。

GROUP=オプションでの分割例

```
/* 乱数からなるサンプルデータの作成 */
DATA RankData2;
  DO i=1 TO 1000;
    x=INT(RANUNI(8901)*700)*10;
    OUTPUT;
  END;
  DROP i;
RUN;
/* GROUP=10 を指定 */
PROC RANK DATA=RankData2 GROUP=10 OUT=RankOut5;
  VAR x;
  RANKS group;
RUN;
/* PRINTプロシジャでデータを確認 */
PROC PRINT DATA=RankOut5 (obs=10);
RUN;
```


出力例

OBS	x	group
1	2460	3
2	5310	7
3	5850	8
4	6350	9
5	1290	1
6	3760	5
7	400	0
8	4110	6
9	1510	2
10	3560	5

その他にも、順位からいわゆる正規スコアの算出を行なうNORMAL=オプションや、Savageスコア(指数スコア)を算出するSAVAGEオプション、順位付けの結果をカスタマイズするFRACTION、NPLUS1オプション、およびPERCENTオプションがPROC RANKステートメントで利用可能です。なお、RANKプロシジャでは度数を表す変数を直接扱うことができません。この件については、弊社WEBページで公開されているFAQをご参照ください。

度数変数を考慮に入れた順位付けについて

<http://www.sas.com/japan/service/technical/faq/list/body/stat072.html>

4.4. データの標準化

指定した平均と標準偏差になるように、ある変数値を標準化(規準化)するためには、Base SASのSTANDARDプロシジャを利用できます。標準化後の平均をMEAN=オプションで、同じく標準偏差をSTD=オプションで指定して下さい。次のプログラムでは、3つの変数x、y、zいずれに対しても、変換後に平均が0、標準偏差が1となるように標準化しています。

STANDARDプロシジャの使用例

```

/* サンプルデータの作成 */
DATA StandardData1;
  INPUT score_x score_y score_z;
DATALINES;
48 61 79
38 25 75
42 83 88
29 12 73
43 75 82
;
RUN;

/* STANDARDプロシジャの使用例 */
PROC STANDARD DATA=StandardData1
  MEAN=0 STD=1 OUT=StandardOut1;
  VAR score_x score_y score_z;
RUN;

/* PRINTプロシジャでデータを確認 */
PROC PRINT DATA=StandardOut1;
RUN;

/* MEANSプロシジャで標準化後の平均と標準偏差を確認 */
PROC MEANS DATA=StandardOut1;
RUN;

```

いわゆる「偏差値」は、平均が50、標準偏差が10となるように標準化することによって算出されます。したがって、次のような指定で計算できます。

MEAN=50,STD=10の例

```

PROC STANDARD DATA=StandardData1
  OUT=StandardOut2 MEAN=50 STD=10;
  VAR score_x score_y score_z;
RUN;

```

もし欠損値がデータに含まれている場合には、標準化の際に使用されずに欠損値のまま出力されます。また、REPLACEオプションをPROC STANDARDステートメントで指定すると、欠損値はその変数の平均で置き換えられます。

REPLACEオプションの使用例

```

/* サンプルデータの作成 */
DATA StandardData2;
  INPUT x y z;
DATALINES;
4.0 1.2 2.9
2.5 . . <--- 変数yとzは欠損値
3.1 1.8 6.7
4.2 2.2 8.2
3.4 . 1.3 <--- 変数yは欠損値
;
RUN;

/* STANDARDプロシジャの使用例 */
PROC STANDARD DATA=StandardData2
  REPLACE OUT=StandardOut3;
  VAR x y z;
RUN;

/* PRINTプロシジャでデータを確認 */
PROC PRINT DATA=StandardOut3;
RUN;

```

出力例

OBS	x	y	z
1	4.0	1.20000	2.900
2	2.5	1.73333	4.775
3	3.1	1.80000	6.700
4	4.2	2.20000	8.200
5	3.4	1.73333	1.300

なお、REPLACEオプションを指定しているときにMEAN=、またはSTD=オプションを併せて指定すると、欠損値をもとの変数の平均で置き換えられた後に標準化が行なわれます。

REPLACEオプションとMEAN=,STD=オプションの使用例

```
PROC STANDARD DATA=StandardData2
  REPLACE MEAN=0 STD=1 OUT=StandardOut4;
RUN;
```

標準化、と一口に言っても、平均が0、標準偏差が1に変換することだけに実は限りません。たとえば、もとの変数値を最小値が0、最大値が1となるように変換する方法なども含まれます。

SAS8以降では、SAS/STATに追加されたSTDIZEプロシジャを使用して、このような広義の意味での標準化を直接扱うことができます。

前記のケースについては、以下のようにMETHOD=RANGEという指定をPROC STDIZEステートメントで指定します。

STDIZEプロシジャの使用例

```
PROC STDIZE DATA=StandardData1
  METHOD=RANGE out=StdizeOut1;
  VAR score_x score_y score_z;
RUN;
      /* PRINTプロシジャでデータを確認 */
PROC PRINT DATA=StdizeOut1;
RUN;
```

出力例:各変数の最大値は1、最小値は0

OBS	score_x	score_y	score_z
1	1.00000	0.69014	0.40000
2	0.47368	0.18310	0.13333
3	0.68421	1.00000	1.00000
4	0.00000	0.00000	0.00000
5	0.73684	0.88732	0.60000

STDIZEプロシジャには、METHOD=オプションで指定できる標準化のキーワードは18個存在します。もとの変数値をoriginal、計算される変換後の値をresultとすると、いずれの方法も

$$\text{result} = \text{add} + \text{multiply} \times [(\text{original} - \text{location}) / \text{scale}]$$

という計算式に従って変換が行われます。なお、addとmultiplyはそれぞれADD=、MULT=オプションで指定可能であり、デフォルトではadd=0、multiply=1となっていて、変換そのものに影響を与えていません。

代表的な変換方法

METHOD=で指定するキーワード	"location"	"scale"	備考
MEAN	平均	1	平均が0となるようにシフト
MEDIAN	中央値	1	中央値が0となるようにシフト
SUM	0	合計	変換後の合計は1

オプション	変換後の値	標準偏差	STANDARDプロシジャと同等
STD (デフォルト)	平均	標準偏差	STANDARDプロシジャと同等
RANGE	最小値	最大値-最小値	変換後の範囲は0 ~ +1
MIDRANGE	(最大値+最小値)/2	(最大値-最小値)/2	変換後の範囲は-1 ~ +1

また、標準化における4つのパラメータ"location" "scale" "add" "multiply"をSASデータセットとして与え、STDIZEプロシジャに読み込ませて変換を行なう機能もあります。なお、STDIZEプロシジャにおいても、欠損値を数値で置き換える機能があります。いくつかの例が、下記のWEBサイトで公開されています。

単一代入法や多重代入法による欠損値への代入方法

<http://www.sas.com/japan/service/technical/faq/list/body/stat063.html>

4.5. 相関係数の計算、相関分析

相関係数を計算する場合には、CORRプロシジャの利用がまず考えられます。

CORRプロシジャは、よく使われる4つの相関係数を出力できます。

- ・ Pearson (ピアソン) の積率相関係数
- ・ Spearman (スピアマン) の順位相関係数
- ・ Kendall (ケンドール) の順位相関係数 (τ-b)
- ・ Hoeffding (ヘフディング) のD統計量

これらのうちPearsonの積率相関係数だけを出力するのであれば、特別の指定は必要ありません。

CORRプロシジャの使用例

```
      /* 乱数からなるサンプルデータの作成 */
DATA CorrData1;
  DO i=1 to 100;
    x=RANUNI(123)*10;
    y=x+RANNOR(123)*2+10;
    z=y+RANNOR(123)*100+20;
  OUTPUT;
  END;
RUN;
      /* CORRプロシジャ */
PROC CORR DATA=CorrData1;
  VAR x y z;
RUN;
```

もし複数のタイプの相関係数を計算するのであれば、それぞれPEARSON、SPEARMAN、KENDALL、HOEFFDINGの各オプションをPROC CORRステートメントで指定してください。なお、相関係数ではありませんが、信頼性係数(クロンバックのα)は、ALPHAオプションを指定して得られます。

CORRプロシジャ (PEARSON SPEARMANオプション指定) の使用例

```
PROC CORR DATA=CorrData1 PEARSON SPEARMAN;
  VAR x y z;
RUN;
```

出力例

```

CORR プロシジャ

3 変数 :      x      y      z

      要約統計量

変数      N      平均      標準偏差      中央値

x      100      5.49662      2.64695      5.56634
y      100      15.36071      3.11228      15.92239
z      100      34.92645      83.37819      20.70364

      要約統計量

変数      最小値      最大値

x      0.01092      9.99491
y      7.38125      20.61908
z     -128.28164      263.07416

Pearson の相関係数, N = 100
H0: Rho=0 に対する Prob > |r|

      x      y      z

x      1.00000      0.81160      0.15076
      <.0001      0.1343
y      0.81160      1.00000      0.18981
      <.0001      0.0586
z      0.15076      0.18981      1.00000
      0.1343      0.0586

Spearman の相関係数, N = 100
H0: Rho=0 に対する Prob > |r|

      x      y      z

x      1.00000      0.80196      0.14418
      <.0001      0.1524
y      0.80196      1.00000      0.18260
      <.0001      0.0690
z      0.14418      0.18260      1.00000
      0.1524      0.0690
    
```

この出力は、大きく3つの部分、すなわち「変数の情報」、「要約統計量」、「相関係数」に分かれています。最後の相関係数のテーブルでは、上段に相関係数そのものが、下段に母相関係数=0を帰無仮説とした両側検定におけるp値が出力されます。たとえば、xとzのPearsonの相関係数は0.15076であり、検定の結果のp値は0.1343と報告されています。

CORRプロシジャにおける欠損値の扱いは、他の多くの分析系プロシジャとは異なります。デフォルトでは、欠損値は「ペアワイズ」に削除されます。言い換えると、なるべく多くのデータを使用して相関係数が算出されます。

CORRプロシジャにおける欠損値の扱い

```

/* サンプルデータ */

X1  X2  X3
2.3 2.6 5.7
4.5 1.4 .
1.2 4.1 8.6
5.7 . .
5.3 5.1 2.5
1.3 2.4 5.4
    
```

相関係数が計算される際には、以下のデータが使用されます。

X1	X2	X2	X3	X3	X1
2.3	2.6	2.6	5.7	5.7	2.3
4.5	1.4	4.1	8.6	8.6	1.2
1.2	4.1	5.1	2.5	2.5	5.3
5.3	5.1	2.4	5.4	5.4	1.3
1.3	2.4				

ただし注意すべき点として、このように計算された相関係数行列や分散共分散行列は、いわゆる正の定符号 (positive definite) 行列とはならないことがあります。これに対して、NOMISSオプションをPROC CORRステートメントで指定すると、他の多くのプロシジャと同様に欠損値を「リストワイズ」に削除、すなわち行単位でオブザベーションが削除されます。

NOMISSオプションを利用したときの欠損値の扱い

以下のデータが計算で使用されます。

```

/* サンプルデータ */

X1  X2  X3
2.3 2.6 5.7
1.2 4.1 8.6
5.3 5.1 2.5
1.3 2.4 5.4
    
```

CORRプロシジャには他にも数多くの機能が存在します。

たとえば、アウトプットへの出力を制御するオプションをPROC CORRステートメントで指定できます。

NOSIMPLE	要約統計量のテーブルを出力しない。
NOCORR	Pearsonの相関係数に関するテーブルを出力しない。
NOPROB	要約統計量のテーブルのうち、母相関係数=0の検定におけるp値を出力しない。
NOPRINT	全ての出力を行わない。OUTP=オプションなどの利用が前提。

また、VARステートメントで変数を指定すると、それらの全ての組み合わせについて相関係数が計算されます。前記の例では合計9個の相関係数が出力されますが、それら全てを必要としないこともあるでしょう。このようなときには、WITHステートメントが有用です。たとえば、変数xとy、および変数xとzの相関係数を求めるには、以下のように指定します。

WITHステートメントの使用例

```
PROC CORR DATA=CorrData1;
  VAR y z;
  WITH x;
RUN;
```

一般にVARステートメント、およびWITHステートメントで指定した変数の個数をそれぞれm、nとすると、m×n個の相関係数が得られます。

その他の有用な機能として、以下のものがあります。

- ・ 偏相関係数の算出。PARTIALステートメントで、影響を除去したい変数を指定します。
- ・ Fisherのz変換に基づく母相関係数に関する検定。PROC CORRステートメントでFISHERオプションを使用します。
- ・ 共分散行列の出力。COVオプションをPROC CORRステートメントで指定します。

「相関」という名前のつく統計量は、他にもたくさん存在します。CORRプロシジャでサポートされていないある種の「相関」は、他の方法で求めることができます。

- ・ Polychoric (ポリコリック) 相関やtetrachoric (テトラコリック) 相関、Stuart (スチュアート) の τ -cは、FREQプロシジャで算出されます。
- ・ 一連の双列 (biserial) 相関については%BISERIALマクロが提供されています。

Compute biserial, point biserial, and rank biserial correlations
<http://support.sas.com/ctx/samples/index.jsp?sid=490>

- ・ 級内 (intraclass) 相関係数は、%INTRACCLASSマクロが用意されています。その他、SAS/STATのNESTEDプロシジャを利用してある種の級内相関係数を求めることもできます。

Compute six intraclass correlation measures
<http://support.sas.com/ctx/samples/index.jsp?sid=537>

5. SAS® Enterprise Guide® での対応

これまでにご紹介したプロシジャの中で、STDIZEプロシジャ以外については、Windows上のGUIアプリケーションであるSAS Enterprise Guideにおいてメニューから選べるタスクとして用意されています。

	EG4.1	EG3
MEANS	記述統計>「要約統計量」タスク	←
UNIVARIATE	記述統計>「分布」タスク	分析>「分布」タスク
RANK	データ>「ランク」タスク	←
STANDARD	データ>「データの標準化」タスク	←
CORR	分析>多変量解析>「相関関係」タスク	分析>「相関関係」タスク

6. おわりに

様々なアプリケーションがSASでは現在提供されていますが、その内部ではこれまでにご紹介したようなSASの分析プロシジャが実行されています。そのため、このようにプロシジャの利用方法を知っておくと、色々な場面で有用になることでしょう。今後のTechnical Newsでも、定期的にご紹介してまいります。



Q&A



- ODS RTFのページ番号表示を調整する
- クライアントユーザーとしてインストールした場合のヘルプの参照
- SASのプログラムから外部ディレクトリを作成する
- SASプログラムファイルを読み取り専用に変更する
- 複引用符に囲まれたカンマ区切りのデータの読み込みについて
- Enterprise Guide 3.0で標準の出力ライブラリを変更する方法
- TABULATEプロシジャ出力の列幅指定方法について
- GPLOTプロシジャで一部の凡例のみを表示
- SAS/IMLでの行列における欠損値の置き換え
- 線形回帰におけるAICに基づく変数選択
- EXPANDプロシジャを利用した前のオペレーションの欠損値ではない値のセット方法

Q ODSを使用し、RTF形式で複数ページの出力結果を得る際に、ページ番号を「1 of n」、「2 of n」のように表示することはできますか。

A ODSの”In-line Formatting Function”を利用することで、ページ番号表記を編集することが可能です。SAS9では”|thispage|” Functionと、”|lastpage|” Functionを利用して、たとえばタイトル行やフットノートなどに、簡易な定義のみで出力することが可能です。また、ODSのPDF形式での出力においても、同じ指定で対応することができます。SAS 8.2では、PDF出力には対応していませんが、RTF形式の出力では”In-line Formatting Function”として、直接”RTFの制御コード”を記述することにより、ページ番号表記を編集することが可能です。

※ODSの”In-line formatting function”は、SAS 8.2,SAS 9.1ではExperimentalの機能です。

以下の例では、ページ番号をタイトル行に右寄せで表示していますが、フットノートでも指定できます。また、左寄せや中央寄せにすることもできます。行揃えは、TITLEステートメント、FOOTNOTEステートメントでJ (JUSTIFY) オプションを指定して制御できます。

例:ODS RTFでの指定例 (SAS 9.1)

```
ODS ESCAPECHAR='^';
ODS RTF FILE='c:\temp\test.rtf';
OPTIONS NONUMBER;
TITLE 'タイトル' J=R 'Page ^{thispage} of ^{lastpage}';
PROC PRINT DATA=sashelp.air;
RUN;
ODS RTF CLOSE;
```

例:ODS RTFでの指定例 (SAS 8.2およびSAS 9.1)

```
ODS RTF FILE='c:\temp\test2.rtf';
OPTIONS NONUMBER;
TITLE 'タイトル' J=R
"#{field{**fldinst {#b#i PAGE }}#~{#b#i of}#~{field{**fldinst
{#b#i NUMPAGES }}}";
PROC PRINT DATA=sashelp.air;
RUN;
ODS RTF CLOSE;
```

Q Windows版SAS 9.1をクライアントユーザーモデルとしてインストールしましたが、ヘルプを参照しようとしてもページが表示されません。どのようにすれば参照可能になりますか。

A クライアントユーザーとしてSASをインストールした場合(※)、SASのプログラムイメージが保存されている、サーバー上のヘルプファイル(拡張子.chm)が参照されます。ファイルサーバーなどの参照設定など、ご利用の環境によってはネットワーク上の.chmファイルを開けないことがあります。その場合は次の手順で作業を実施し、ヘルプが参照可能となるか試してみてください。

[ご注意] 以下の作業はWindowsのレジストリ操作を伴います。事前にバックアップを取得しておくことを推奨します。

1. [スタート] > [ファイル名を指定して実行]にて、regedit と入力し、レジストリエディタを起動します。
2. 次のレジストリエントリを検索します。

```
HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\HTMLhelp\1.x\
```

3. 新規キーを作成し、名称をItssRestrictionsとします。
4. ItssRestrictionsにて、新規”文字列値”を作成し、名称をUrlAllowListとします。
5. UrlAllowListの値に、.chmファイルが存在するサーバー上のフォルダを指定します。以下は例となります。実際の環境により読み替えてください。

例) ネットワーク上の次のパスにクライアントユーザー用のイメージが作成されているものとします。

```
\\Server\SAS Server\9.1
```

この場合、次の場所に.chmファイルが存在します。

```
\\Server\SAS Server\9.1\sas\nls\ja\help
\\Server\SAS Server\9.1\sas\core\help
```

UrlAllowListへの指定値は次のようになります。実際には改行を含めずに1行で記述します。

```
\\Server\SAS Server\9.1\sas\nls\ja\help%;
file://\\Server\SAS Server\9.1\sas\nls\ja\help;
\\Server\SAS Server\9.1\sas\core\help%;
file://\\Server\SAS Server\9.1\sas\core\help
```

6. レジストリエディタを閉じます。

(※)このインストール形態についての詳細は「Microsoft Windows版 SAS 9.1.3 Foundation 管理者ガイド」をご参照ください。

Q SASシステムのプログラムから、オペレーティングシステム上へ直接新しいディレクトリを作成する方法はありますか。

A SAS9以前では、Xステートメントを使用して、SASシステムからオペレーティングシステムのコマンドを発行することでディレクトリを作成することができました。SAS9からは、新たに追加されたDCREATE関数を使用することで、Xステートメントを使用せずとも外部ディレクトリを作成することができます。DCREATE関数を使用する利点としては、指定したディレクトリが正常に追加できたか、またはエラーになったかを%SYSFUNCでマクロ関数から処理のリターンコードが受けとることができるので、プログラム中でエラーハンドリングが可能です。

DCREATE関数の構文

```
DCREATE (ディレクトリ名, 親ディレクトリ名);
```

説明

ディレクトリ名	作成したいディレクトリの名称
親ディレクトリ名	作成するディレクトリの上位ディレクトリ

例:Xステートメントの使用例

```
OPTIONS NOXWAIT;
X 'mkdir C:¥TEMP¥testdir';
```

例:DCREATE関数の使用例

```
%LET rc=%SYSFUNC(DCREATE(testdir, C:¥TEMP¥));
```

Q SASプログラムファイルを誤って更新しないために、読み取り専用にする方法はありませんか。

A SASシステムには、作成したプログラムファイルの属性を変更する機能はありません。必要に応じて、OS側の機能を使用することになります。以下の例では、SASからOSのコマンドを実行するXステートメントを使用しています。Xステートメントの引数では、ファイル属性を変更する為のWindowsコマンドである"attrib" コマンドを実行し、C:¥test.sasを読み取り専用にしています。

例:Xステートメントの使用例

```
OPTIONS NOXWAIT;
X 'attrib +R "C:¥test.sas";
```

Q 複引用符に囲まれたカンマ区切りのデータがあります。複引用符をデータとして読み込むには、どのようにすればよいでしょうか。

A カンマ区切りのデータを読み込むには、INFILEステートメントにDSDオプションを利用します。このDSDオプションでは、複引用符をデータとして扱いません。複引用符がデータとして必要な場合は、" ~ (チルダ) フォーマットモディファイヤ" をINPUTステートメントに追加することで対応可能です。

以下の例では、aa,bb,ccの3変数からなるカンマ区切りのデータを読み込み、bbのデータに対して複引用符をそのままデータとして読み込んでいます。

例:複引用符つき、カンマ区切りデータを読み込む例

```
DATA data1;
  INFILE DATALINES DSD;
  LENGTH aa bb cc $10;
  INPUT aa bb ~ cc;
CARDS;
"DATA1", "DATA2", "DA, TA3"
"DATA4", , "DATA5"
;
RUN;
```

Q Enterprise Guide 3.0において、タスクやクエリを実行した際に、SASUSERライブラリが出力先として選択されます。この出力先を変更することは可能ですか。

A 以下の手順にて出力ライブラリのオプション値に利用可能なライブラリを追加指定することで、標準の出力先を変更することが可能です。

[ツール]メニュー → [オプション] → [タスク] → [出力ライブラリ]

1. [ツール] > [オプション] とたどり、オプションメニューを開きます。
2. [タスク] > [出力ライブラリ] を選択し、「出力データのデフォルトライブラリ」にて、「追加」ボタンを押し、任意のライブラリ名を指定します。
3. 追加したライブラリが一番上に表示されるように、順序を変更します。
4. OKボタンを押して、オプションメニューを終了します。

※利用可能なライブラリについては、接続先のSASサーバーでの設定により異なります。

Q TABULATEプロシージャにて、ODSのHTMLやPDF出力の際の列幅を指定するにはどのようにしたらよいでしょうか。

A STYLE=オプションのCELLWIDTH=オプションを指定することにより、該当変数の列幅を指定することが可能です。

STYLE=オプションは、VARステートメント、CLASSステートメントやTABLESステートメント等で指定することができます。次ページの例を参考にしてください。

例:TABULATEプロシジャにてODS HTML出力でのSTYLE=オプション指定例

```
ODS HTML FILE='C:\temp\test1.html';
PROC TABULATE DATA=sashelp.class;
  CLASS sex / STYLE=[CELLWIDTH= 5 cm];
  VAR weight / STYLE=[CELLWIDTH= 3 cm];
  VAR height ;
  TABLES sex, (weight height*[STYLE=[CELLWIDTH=5 cm]]);
RUN;
ODS HTML CLOSE;
```

また、CLASSLEVステートメントを使用して、CELLWIDTHオプションとフォーマット(出力形式)を組み合わせることにより、列を形成する変数の値ごとに列幅を変更することも可能です。次の例では、ユーザー定義フォーマット\$sexw.を用意し、CLASSLEVステートメントにて指定し、変数sexの値ごとに出力幅を変更しています。

例:CLASSLEVステートメントでユーザー定義フォーマットを利用する指定例

```
PROC FORMAT;          /* 変数 SEX に対するフォーマット作成 */
VALUE $sexw F='2 cm' /* F の場合 幅2cm M の場合 1cm */
          M='1 cm';
RUN;
ODS HTML FILE="C:\temp\test2.html";
PROC TABULATE DATA=sashelp.class;
  CLASS sex age;
  CLASSLEV sex / STYLE=[CELLWIDTH=$sexw.];
                                     /*フォーマットを使用*/
  CLASSLEV age / STYLE=[CELLWIDTH=5 cm];
  TABLE age, sex;
RUN;
ODS HTML CLOSE;
```

Q GPLOTプロシジャにて、複数の応答変数(Y軸)を用い、OVERLAYオプションを指定し、グラフを作成しています。この際、LEGENDオプションを追記すると全ての凡例が表示されます。一部の凡例のみを表示することはできますか。

A GPLOTプロシジャでは、PLOTステートメントのほか、PLOT2ステートメントをあわせて指定することができます。この際、片方のステートメントにのみLEGENDオプションを指定することによって、一部の凡例のみを表示させることが可能です。

たとえば、各X軸の点において、エラーバーのグラフを描いた後、平均値を任意のシンボルにて表し、値の変移を描くとします。このため、データセットには、応答変数Y1,Y2、変数Xに加え、それぞれの平均値を表すMEAN1,MEAN2が含まれているとします。この場合、右記のように記述することによって、平均値の偏移のみに対する凡例を表示させることができます。

例:PLOT2ステートメントを利用したLEGENDオプションの振り分け例

```
                                     /* Y軸を設定 */
AXIS1 ORDER=(-2 TO 10 BY 2);
PROC GPLOT DATA=test;
                                     /* エラーバー部分のグラフを作成 */
PLOT (y1 y2)*x / OVERLAY VAXIS=AXIS1;
  SYMBOL1 I=std V=none C=red;
  SYMBOL2 I=std V=none C=blue;
                                     /* 平均値偏移部分のグラフを作成 */
                                     /* LEGENDオプションを指定 */
PLOT2 (mean1 mean2)*x / OVERLAY VAXIS=AXIS1 LEGEND;
  SYMBOL3 I=join V=dot C=red;
  SYMBOL4 I=join V=dot C=blue;
RUN;
QUIT;
```

※PLOT、PLOT2ステートメントにて同じ設定のY軸とするため、明示的にAXISステートメントを用いています。

Q SAS/IMLにて、行列に含まれる欠損値を任意の値に置き換えるにはどのようにしたらいいですか。

A 行列の要素を置き換える関数として、CHOOSE関数があります。この関数では、最初の引数にて条件を指定し、その条件が真である場合には2番目の引数、真でない場合には3番目の引数とする行列を作成することができます。このため、行列Aにおける欠損値を0に置き換える場合には、以下のように指定することにより可能です。

例:CHOOSE関数の指定例

```
a=CHOOSE (a=., 0, a);
```

Q 赤池の情報量規準(AIC)を使用した変数選択をREGプロシジャで実現することは可能でしょうか。

A REGプロシジャではいくつかの変数選択法が用意されています。たとえば、SELECTION=ADJRSQL=MODELステートメントのオプションとして指定すると、調整済みR2乗値(調整済み寄与率)の大きい順にモデルがアウトプットへ出力されます。

AICに基づく変数選択法は直接はサポートされていませんが、線形回帰モデルにおいて説明変数(独立変数)の個数を固定して考えた上で、R2乗値(寄与率)の大きな順でモデルを並べた場合と、AICが小さい順にモデルを並べた場合を比較すると、それらの統計量の計算式から並び順は一致します。この事実を応用すれば、たとえばAICに基づいて上位10個のモデルを選ぶときには、右記のようなプログラムで実質的にAICに基づく変数選択を行なうことができます。

例:REGプロシジャでのAICに基づく変数選択の例

```

/* モデルの関数を指定 */
%LET num=10;
ODS LISTING CLOSE;
ODS OUTPUT SubsetSelSummary=SummaryOut;
PROC REG DATA=test;
  MODEL y=x1-x8 / SELECTION=RSQUARE AIC BEST=&num;
RUN;
QUIT;
ODS LISTING;

/* AIC でソート */
PROC SORT DATA=SummaryOut OUT=SortSummaryOut;
  BY AIC;
RUN;

/* アウトプット画面に出力 */
PROC PRINT DATA=SortSummaryOut (OBS=&num)
  LABEL NOOBS;
  VAR NumInModel Rsquare AIC VarsInModel;
RUN;
    
```

例:出力例

取り込んだ 変数の数	R2 乗	AIC	モデルの独立変数
5	0.6064	-66.3777	x1 x2 x4 x7 x8
6	0.6071	-66.2663	x1 x2 x3 x4 x7 x8
6	0.6067	-65.2617	x1 x2 x4 x5 x7 x8
7	0.6075	-65.2402	x1 x2 x3 x4 x5 x7 x8
6	0.6064	-64.3902	x1 x2 x4 x6 x7 x8
7	0.6071	-64.2811	x1 x2 x3 x4 x6 x7 x8
7	0.6067	-63.2676	x1 x2 x4 x5 x6 x7 x8
8	0.6075	-63.2475	x1 x2 x3 x4 x5 x6 x7 x8
4	0.5960	-42.4039	x1 x2 x4 x7
5	0.5968	-42.3157	x1 x2 x3 x4 x7



任意の変数が欠損値のとき、欠損値の代わりに前のオブザベーションの値 (非欠損値) をセットする方法はありますか。

データのイメージ

NO	VALUE
1001	100
.	50
1002	50
.	100
.	150
1003	50
.	200
1004	50

取得したいイメージ

NO	VALUE
1001	100
1001	50
1002	50
1002	100
1002	150
1003	50
1003	200
1004	50



弊社WEBページでは、次のURLにあるようなDATAステップによる方法をご紹介します。

任意変数が欠損値のとき、前のオブザベーションの欠損値ではない値のセット方法
<http://www.sas.com/japan/service/technical/faq/list/body/ba160.html>

上記にあるようなDATAステップによる方法もありますが、ここではSAS/ETSのEXPANDプロシジャを使用して実現するプログラムをご紹介します。次のプログラムは、データセットsampleを使用して、欠損値を置き換えたデータセットoutを作成しています。METHOD=STEPの指定によって、直近の非欠損値をもって欠損値を置き換えます。

例:EXPANDプロシジャによる、欠損値補完の例

```

/* サンプルデータセット作成 */
DATA sample;
  INPUT no value;
CARDS;
1001 100
. 50
1002 50
. 100
. 150
1003 50
. 200
1004 50
;
RUN;

/* EXPANDプロシジャ */
PROC EXPAND DATA=sample OUT=out (DROP=time);
  CONVERT no / METHOD=STEP;
RUN;
    
```


New Publications

新刊マニュアルのお知らせ

「The Essential Guide to SAS® Dates and Times」

http://support.sas.com/publishing/bbu/authorline/2006/6_06/6_06_book.html

注文番号: 59411

I S B N: 1-59047-884-3

価 格: 5,775円 (税込)

本書は、多くのユーザーがつかずきやすい、SASで使用している日付、時間、日時について、その基本概念を説明する総合ガイドです。すべての日付、時間、日時に関する入出力形式が実例つきで1ページにまとめられた便利な「クイック・リファレンスガイド」付きです。

「SAS® for Mixed Models, Second Edition」

http://support.sas.com/publishing/bbu/companion_site/59882.html

SAS®9にあわせて全面改訂されたSASを使用した混合モデルのための最新必携ガイドです。大変便利なCD-ROM版SASコード集付きです。

注文番号: 59882

I S B N: 1-59047-500-3

価 格: 17,220円 (税込)

「SAS® Certification Prep Guide: Base Programming for SAS®9」

<http://support.sas.com/publishing/pubcat/tocs/61034.html>

今夏にリリース予定のSAS®9対応の認定試験 SAS Base Programming Exam for SAS®9の試験対策用公式ガイドです。付属のCD-ROMにはチュートリアル、練習用データ、サンプルプログラムを収録しています。

注文番号: 61034

I S B N: 1-59047-922-X

価 格: 23,730円 (税込)

■8月の新刊発売予定

「The Little SAS® Book for Enterprise Guide® 4.1」

著 者: Susan J. Slaughter and Lora D. Delwiche

「Web Development with SAS® by Example, Second Edition」

著 者: By Frederick E. Pratter, Ph.D.

■SAS Publication Catalog 2006 (抜粋版) 印刷版のご案内

SAS Publication Catalog 2006 (抜粋版) 印刷版が完成しました。

ご希望の方はご住所・お名前・電話番号を明記の上

JPNBookSale@sas.comまで電子メールでお申し込みください。

SASマニュアル申込用紙、および最新のPublication Catalog (マニュアル案内パンフレット) は弊社ホームページ (<http://www.sas.com/japan/manual/>) にて公開しておりますので、併せてご利用ください。

マニュアル販売係

- T E L 03-3533-3835
- F A X 03-3533-3781
- E-mail JPNBooksale@sas.com

SAS Training

SASトレーニングのお知らせ

■ 特別トレーニングコース開催のご案内

● 「医薬向けポアソン回帰」コース

日 程：2006年10月13日(金) 10:00～17:00 (東京会場)

価 格：50,400円(税込) / チケット捺印数1

受講対象：SASによる統計解析に従事する方

前提知識：「医薬向けカテゴリカルデータ解析2」コースを受講済みか、同程度の知識のある方

学習内容：本コースでは、計数データに対するポアソン回帰モデルの当てはめ、発生率に対するポアソン回帰モデルの当てはめ、過大分散 (overdispersion) に関するモデル評価、負の二項回帰モデルの当てはめ、モデル診断について学びます。

※テキストは英語版を使用します。

コンテンツ：■ポアソン回帰モデル

- ・ポアソン回帰入門
- ・過大分散の補正

■ポアソン回帰モデルの応用

- ・発生率に対するポアソン回帰モデル
- ・モデル診断

担当講師：東京大学大学院 医学系研究科 生物統計学教室

■ ディスカウント制度一部終了のご案内

ディスカウント制度の「セットプラン」、「グループ受講割引」、および「早期お申し込み割引」の3種を、2006年9月30日のお申し込みをもって受付終了とさせていただきます。

SAS Institute Japan株式会社では、今後も多岐にわたったトレーニングコースを追加していく予定です。コース内容・日程等の詳細は、順次弊社Webサイトに公開しますので、以下のURLをご参照ください。

<http://www.sas.com/japan/training/>

その他、トレーニングに関する情報については、上記のURLをご参照いただくか、下記トレーニング担当までお問い合わせください。

トレーニング担当

- TEL 03-3533-3835
- FAX 03-3533-3781
- E-mail JPNTraining@sas.com

SAS Certification Information

SAS認定試験のお知らせ

■ 認定試験のご案内

今夏SAS®9対応の認定試験 SAS Base Programming Exam for SAS®9 がリリースされます。1年以上のSASプログラミング経験があり、SAS®9を用いて以下の処理が出来る方に受験をお勧めします。

- ・ローデータファイルのインポートとエクスポート
- ・データの処理と変換
- ・SASデータセットの結合
- ・SASプロシジャを用いた基本的な詳細と要約レポートの作成
- ・構文とプログラミングエラーの発見と修正

新しい試験の詳細は下記URLをご覧ください。

<http://www.sas.com/japan/training/certify/index.html>

8月よりこのSAS Base Programming 認定試験受験を目指す方々を対象に新しいトレーニング「SAS Base Programming 認定試験受験対策講座」を開催いたします。

● 「SAS Base Programming 認定試験受験対策講座」

日 程：2006年8月30日(水)～31日(木) 10:00～17:00 (東京会場)

価 格：210,000円(税込)

受講対象：SAS Base Programming 認定試験の合格を目指している方
前提知識：「SASプログラミング I」および「SASプログラミング II」コースを受講済みの方で、SASプログラミングの経験1年以上、もしくは同程度の技術をお持ちの方

学習内容：このコースでは、SAS Base Programming 認定試験の概要、傾向と対策を熟知し、模擬試験や演習を通してSAS Base Programming 認定試験の合格を目指します。

コンテンツ：・試験概要、傾向と対策

- ・講義
- ・模擬試験、自己採点
- ・模擬試験解答

詳しくは下記URLをご覧ください。

<http://www.sas.com/japan/training/index.html>

その他、SAS認定試験に関する情報については、下記SAS認定試験担当までお問い合わせください。

SAS認定試験担当

- TEL 03-3533-3835
- FAX 03-3533-3781
- E-mail JPNCertification@sas.com

Latest Releases

最新リリース情報

■PCプラットフォーム

Windows版	SAS 9.1.3	9.1 TS1M3
64-bit Windows (Itanium) 版	SAS 9.1.3	9.1 TS1M3

■UNIXプラットフォーム

Tru64版	SAS 9.1.3	9.1 TS1M3
SunOS/Solaris版	SAS 9.1.3	9.1 TS1M3
HP-UX版	SAS 9.1.3	9.1 TS1M3
HP-UX (Itanium) 版	SAS 9.1.3	9.1 TS1M3
AIX版	SAS 9.1.3	9.1 TS1M3
Linux (Intel) 版	SAS 9.1.3	9.1 TS1M3
ABI+版	SAS 6.11	TS040

■ミニコンピュータプラットフォーム

OpenVMS AXP版	SAS 6.12	TS020
OpenVMS VAX版	SAS 6.08	TS407

■メインフレームプラットフォーム

IBM版 (OS/390, z/OS)	SAS 9.1.3	9.1 TS1M3
富士通版 (F4, MSP)	SAS 6.09E	TS470
日立版 (VOS3)	SAS 6.09E	TS470
CMS版	SAS 6.08	TS410

Information

SAS Technical News 送付についてのご案内

SAS Technical Newsは次の方を対象にお送りしています。

- ・ SASコンサルタントとしてご登録の方
- ・ SAS Technical Newsの購読をお申し込みいただいている方

今後SAS Technical News購読が不要の方、配信先の変更等をご希望の方は、下記URLよりお手続きください。

配信停止

<http://www.sas.com/japan/corporate/material.html>

配信先変更手続き

http://www.sas.com/japan/sasj_privacy.html#inquiry

SAS Technical News Summer 2006

発行
SAS Institute Japan株式会社

テクニカルニュースに関するお問い合わせ先

テクニカルサポートグループ

TEL: 03-3533-3877

FAX: 03-3533-3781

E-mail: JPNTechnews@sas.com



SAS Institute Japan株式会社

www.sas.com/japan/

東京本社
〒104-0054
東京都中央区勝どき1-13-1
イヌイビル・カチドキ
Tel 03 (3533) 6921
Fax 03 (3533) 6927

大阪支店
〒530-0004
大阪市北区堂島浜1-4-16
アクア堂島西館 12F
Tel 06 (6345) 5700
Fax 06 (6345) 5655